



/*

Table 1

*/



local nametable = "tables/rev_landuse.tex"





*-----
eststo clear
local column_number = 1
local column_name = "column" + "`column_number'"
*---	


use "public_data/g30.dta", clear


forvalues xd=1(1)2{
	
	if `xd'==1{
		local xdist = 50
	}
	if `xd'==2{
		local xdist = 25
	}

		use "public_data/g30.dta", clear
		
		drop if adist>`xdist'		

		replace weight 			= 1
	
		gen interdep = careawall>0

		
		*computes mean of dependent var
		egen interm = mean(interdep)
		local vmeandep = interm[1]
		local meandep = string(`vmeandep')
		if index("`meandep'",".")==1{
			local meandep = "0"+substr("`meandep'",1,3)
		}
		if index("`meandep'",".")<5&index("`meandep'",".")>1{
			local meandep = substr("`meandep'",1,4)
		}
		di "`meandep'"

		*computes the number of country pairs
		egen intercp = group(border)
		egen intercpmax = max(intercp)
		local countrypair = intercpmax[1]
		drop intercp*
	
*		xi: areg interdep  varofi   [aw=weight],a(groupfe) cluster(border)
		reghdfe interdep varofi, absorb(groupfe) vce(cluster groupfe country)
		eststo `column_name'
		estadd local space  					" "
		estadd local groupfe 					"Yes"
		estadd local maxdist 					"`xdist'"
		estadd local countrypair 					"`countrypair'"
		estadd local polynomial				"No"
		estadd local nmeandep				"`meandep'"
		local column_number = `column_number' + 1
		local column_name = "column" + "`column_number'"
	

	
*		xi: areg interdep rdist1*    varofi  [aw=weight] ,a(groupfe) cluster(border)
		reghdfe interdep rdist1* varofi, absorb(groupfe) vce(cluster groupfe country)
		eststo `column_name'
		estadd local space  					" "
		estadd local groupfe 					"Yes"
		estadd local maxdist 					"`xdist'"
		estadd local countrypair 					"`countrypair'"
		estadd local polynomial				"Linear"
		estadd local nmeandep				"`meandep'"
		local column_number = `column_number' + 1
		local column_name = "column" + "`column_number'"	
}	
label var interdep  " "	
#delimit ;
esttab column1 column2 column3 column4
using "`nametable'", replace
label
b(3)
se(3)
star(* 0.10 ** 0.05 *** 0.01)
width(1.0\hsize)
nonotes
nolines
noconstant
mgroups(   "\textit{Dep. Var.: Plot is cultivated}", pattern(1 0 0 0)
prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) 
  keep(varofi )
order(varofi )
  stats(groupfe					space	polynomial		space	r2		N					countrypair			space	maxdist							space	nmeandep,
    fmt(0						0		0				0		3		0					0					0		0								0		0)
layout(@						@		@				@		@		@					@					@		@								@		@)	 	
labels("Border x Ethnic FE"		" "		"RD Polynomial"	" "		"R2"	"\# Observations"	"\# Country pairs"	" "		"Max distance to border (km)"	" " 	"Mean Dep. Var.")
 )	
noobs
booktabs
;
#delimit cr


		
